

```

1 ;
2 ;
3 ; Z80 dallamcsengő eredeti vezérlőprogramjának visszafejtése.
4 ; (Ez nincs benne a cikkben.)
5 ;
6 ; Publikálta: Kinczel András, a Rádiótechnika 1987/2 számában. (99.-101. oldalakon)
7 ;
8 ; Szerző: TCSabI
9 ; Dátum: 2017. július 26.
10 ; E-mail: toth.csaba.hip@gmail.com
11 ;
12 ;
13 0000 ED56 IM 1 ; A CPU Interrupt Mode 1-be állítása.
14 ; Az INT bemenétre egy RST 38H
15 ; (azaz CALL 0038H) utasítást fog végrehajtani.
16 ; Az NMI bemenétre egy RST 66H
17 ; (azaz CALL 0066H) utasítást fog végrehajtani.
18 ; Az INT és NMI bemeneteken vannak,
19 ; a dallamindító gombok.
20 ; Visszatárási pont, a dallamok ismételt lejátszához.
21 0002 219F00 LD HL,009FH ; HL = 009FH
22 ; A dallamok kezdőcímének inicializálása.
23 ; (Azért nem 00A0h, mert a HALT után van egy
24 ; INC HL utasítás.)
25 ; Visszatérési pont, a következő dallam lejátszásához.
26 0005 FB EI ; Megszakítások engedélyezése.
27 0006 AF XOR A ; A = A xor A, azaz Akkumulátor nullázása 00h.
28 0007 76 HALT ; CPU működésének felfüggesztése,
29 ; a következő megszakítás kérésig.
30 ; Azaz várakozás, valamelyik gomb megnyomására.
31 ; Visszatérési pont, a következő hang lejátszához.
32 0008 23 INC HL ; HL = HL + 1
33 0009 47 LD B,A ; B = A
34 000a 57 LD D,A ; D = A
35 000b 7E LD A,(HL) ; A = (HL), A dallam hangjának kiolvasása
36 ; a memóriából.
37 000c 5F LD E,A ; E = A, Dallam hangjának mentése E-be.
38 000d E61F AND 1FH ; A = A and 1Fh 00011111 Frekvencia kimaszkolása.
39 000f F680 OR 80H ; A = A or 80h 10000000 Így lesz frekvencia cím.
40 0011 4F LD C,A ; C = A, Így BC regiszterben van a frekvencia.
41 0012 7B LD A,E ; A = E, A dallam hangjának kiolvasása E-ből.
42 0013 07 RLCA ; A eltolása balra 1 bittel.
43 0014 07 RLCA ; A 7.bit -> 0.bitbe töltődik.
44 0015 07 RLCA ;
45 0016 E607 AND 07H ; A = A and 07h 00000111 Hanghossz kimaszkolása.
46 0018 F630 OR 30H ; A = A or 30h 00110000 Így lesz hanghossz cím.
47 001a 5F LD E,A ; E = A, Így a DE regiszterben van a hang hossza.
48 001b 1A LD A,(DE) ; A = (DE), A hang hosszának kiolvasása,
49 ; a táblázatból.
50 001c 57 LD D,A ; D = A
51 001d 0A LD A,(BC) ; A = (BC), A Frekvencia kiolvasása,
52 ; a táblázatból.
53 001e FE00 CP 00H ; A == 00h?, azaz A-00h --> Z bit beállítása.
54 0020 2819 JR Z,003BH ; Relatív ugrás a 003Bh címre, ha Z=1. Szünet...
55 0022 FEFE CP 0FEH ; A == FEh?, azaz A-FEh --> Z bit beállítása.
56 0024 28DF JR Z,0005H ; Relatív ugrás a 0005h címre, ha Z=1.
57 ; Következő dallam lejátszása.
58 0026 FEFF CP 0FFH ; A == FFh?, azaz A-FFh --> Z bit beállítása.
59 0028 28D8 JR Z,0002H ; Relatív ugrás a 0002h címre, ha Z=1.
60 ; Dallamkészlet ismételt lejátszása.
61 002a 4F LD C,A ; C = A
62 002b AF XOR A ; A = A xor A, azaz Akkumulátor nullázása 00h.
63 002c 5F LD E,A ; E = A
64 002d C34600 JP 0046H ; Ugrás a 0046h címre. Hang generálása...
65 ; A hangok hosszúságának táblázata.
66 0030 FF ; Egész
67 0031 C1 ; 3/4
68 0032 81 ; 1/2
69 0033 61 ; 3/8
70 0034 41 ; 1/4
71 0035 31 ; 3/16
72 0036 21 ; 1/8
73 0037 11 ; 1/16

```

```

74 ; A CPU INT bemenetére megszakításkérés érkezett.
75 ; A dallamcsengő INT bementén lévő gombját megnyomták. Dallam lejátszása.
76 0038 C30800 JP 0008H ; Ugrás a 0008h címre. Következő hang lejátszása.
77 ; Hang nélküli várakozás. Azaz szünet lejátszása...
78 003b 0601 LD B,01H ; B = 01h
79 ;*
80 003d 10FE DJNZ 003DH ; B = B - 1, ha B nem lett 0,
81 ; akkor relatív ugrás a 003Dh címre.
82 003f 1B DEC DE ; DE = DE - 1
83 0040 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
84 0041 20F8 JR NZ,003BH ; Relatív ugrás a 003Bh címre, ha Z=0.
85 0043 C30800 JP 0008H ; Ugrás a 0008h címre. Következő hang lejátszása.
86 ; Hang generálása... BC frekvenciával és DE időtartammal.
87 0046 41 LD B,C ; B = C
88 ;
89 0047 1B DEC DE ; DE = DE - 1
90 0048 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
91 0049 28BD JR Z,0008H ; Relatív ugrás a 0008h címre, ha Z=1.
92 ; Következő hang lejátszása.
93 004b 10FA DJNZ 0047H ; B = B - 1, ha B nem lett 0,
94 ; akkor relatív ugrás a 0047h címre.
95 004d D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
96 ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
97 ; A CPU ekkor, a IORQ kimenetére egy túimpulzust
98 ; küld. Az adatbuszon érvényes cím van.
99 ; A túimpulzus, az SN7474-es D tárolóban szim-
100 ; metrikus négyszögjellé alakul. Ezt halljuk.
101 004f C34600 JP 0046H ; Ugrás a 0046h címre.
102 ; A Bim-Bam második hangjának lejátszása...
103 0052 1680 LD D,80H ; D = 80h
104 ;*
105 0054 0624 LD B,24H ; B = 24h
106 ;*
107 0056 1B DEC DE ; DE = DE - 1
108 0057 BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
109 0058 2806 JR Z,0060H ; Relatív ugrás a 0060h címre, ha Z=1.
110 ; Szünet lejátszása.
111 005a 10FA DJNZ 0056H ; B = B - 1, ha B nem lett 0,
112 ; akkor relatív ugrás a 0056h címre.
113 005c D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
114 ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
115 ; A CPU ekkor, a IORQ kimenetére egy túimpulzust
116 ; küld. Az adatbuszon érvényes cím van.
117 ; A túimpulzus, az SN7474-es D tárolóban szim-
118 ; metrikus négyszögjellé alakul. Ezt halljuk.
119 005e 18F4 JR 0054H ; Relatív ugrás a 0054h címre.
120 ; Várakozás...
121 0060 06FF LD B,0FFH ; B = FFh
122 ;*
123 0062 10FE DJNZ 0062H ; B = B - 1, ha B nem lett 0,
124 ; akkor relatív ugrás a 0062h címre.
125 0064 189F JR 0005H ; Relatív ugrás a 0005h címre.
126 ; Visszaugrás a dallam lejátszási üzemmódba.
127 ; A CPU NMI bemenetére megszakításkérés érkezett. A dallamcsengő gombját megnyomták.
128 ; A Bim-Bam első hangjának lejátszása...
129 0066 1620 LD D,20H ; D = 20h
130 ;
131 0068 0620 LD B,20H ; B = 20h
132 ;*
133 006a 1B DEC DE ; DE = DE - 1
134 006b BA CP D ; A == D?, azaz A-D --> Z bit beállítása.
135 006c 2806 JR Z,0074H ; Relatív ugrás a 0074h címre, ha Z=1.
136 ; Szünet lejátszása.
137 006e 10FA DJNZ 006AH ; B = B - 1, ha B nem lett 0,
138 ; akkor relatív ugrás a 0062h címre.
139 0070 D301 OUT (01H),A ; 01h portra adja, az A regiszter tartalmát
140 ; (D0-D7). (A0-A7) = 01h, (A8-A15) változatlan.
141 ; A CPU ekkor, a IORQ kimenetére egy túimpulzust
142 ; küld. Az adatbuszon érvényes cím van.
143 ; A túimpulzus, az SN7474-es D tárolóban szim-
144 ; metrikus négyszögjellé alakul. Ezt halljuk.
145 0072 18F4 JR 0068H ; Relatív ugrás a 0068h címre.
146 ; Várakozás...

```

```

147 0074 06FF      LD      B,0FFH      ; B = FFh
148 ;*
149 0076 10FE      DJNZ   0076H      ; B = B - 1, ha B nem lett 0,
150 ; akkor relatív ugrás a 0076h címre.
151 0078 C35200     JP      0052H      ; Ugrás a 0052h címre.
152 007b FF        ;
153 007c FF        ;
154 007d FF        ; Nem használt terület.
155 007e FF        ;
156 007f FF        ;
157 ; A hangok frekvenciájának táblázat...
158 ; A frekvencia és a várakozások időtartama, a CPU órajelétől függ!
159 0080 00        ; Nincs hang, azaz szünet.
160 0081 11        ; H                +
161 0082 12        ; Aisz            +
162 0083 13        ; A                +
163 0084 14        ; Gisz            +
164 0085 15        ; G                +
165 0086 16        ; Fisz            +
166 0087 18        ; F                +
167 0088 19        ; E                +
168 0089 1B        ; Disz            +
169 008a 1D        ; D                +
170 008b 1E        ; Cisz            +
171 008c 20        ; C                +
172 008d 22        ; H                +
173 008e 24        ; Aisz            előző hang + 2^(1/12) (1 oktáv 12 hang)
174 008f 26        ; A (a normál A hang) 440Hz
175 0090 29        ; Gisz            előző hang - 2^(1/12) (1 oktáv 12 hang)
176 0091 2B        ; G                -
177 0092 2E        ; Fisz            -
178 0093 31        ; E                -
179 0094 34        ; Disz            -
180 0095 36        ; D                -
181 0095 3A        ; Cisz            -
182 0097 3D        ; C                -
183 0098 41        ; H                -
184 0099 45        ; Aisz            -
185 009a 4A        ; A                -
186 009b 4D        ; Gisz            -
187 009c 51        ; G                -
188 009d 56        ; Fisz            -
189 009e FE        ; Dallam vége jelző.
190 009f FF        ; Dallamkészlet vége jelző.
191 ; Innem kezdődnek a dallamok...
192 00A0 .. .. ..
193 ; A tárolt hang szerkezete...
194 ; Egy bájt, 1 hang.
195 ; A hang hossza: 7-5 bitek, azaz 3 biten.
196 ; A hang frekvenciája: 4-0 bitek, azaz 5 biten.

```